

Differentiation of Discrete Functions

Ana Catalina Torres, Autar Kaw
 University of South Florida
 United States of America
 kaw@eng.usf.edu

Introduction

This worksheet demonstrates the use of Mathcad to illustrate the differentiation of discrete functions using:

a) Forward Divided Difference Method to find the derivative of a function given at discrete $n + 1$ data points $(x_1, y_1), (x_2, y_2), \dots, (x_n, y_n)$, the value of the $f'(x)$ for $x_i \leq x \leq x_{i+1}$, $i = 1, \dots, n-1$, is given by

$$f'(x_i) \approx \frac{f(x_{i+1}) - f(x_i)}{x_{i+1} - x_i}$$

- b) By differentiation of a second order interpolated polynomial, and
 c) By differentiation of a polynomial using all data points.

Section 1: Input

The following simulation finds the approximate value of the first derivative of a discrete function using Forward Difference Divided Difference. The user inputs are

- a) data with x and y values. At least 3 data points are required.
 b) point at which the derivative is to be found, x_v

The outputs include

- a) approximate value of the derivative at the given point using Forward Divided Difference
 b) approximate value of the derivative at the given point by differentiating a second order polynomial found using the three closest values to x_v , and that bracket x_v .
 c) approximate value of the derivative at the given point using all data points and differentiating a $n-1^{th}$ order polynomial that goes through the n data points

Data points, y vs. x

$x := (10 \ 15 \ 20 \ 21 \ 25)$

$y := (100 \ 225 \ 400 \ 441 \ 625)$

Value of x at which $f'(x)$ is desired, x_v

$x_v := 11$

This is the end of the user section. All the information must be entered before proceeding to the next section.

Section 2: Calculation

The following loop estimates the first derivate of a discrete function at a point xv using Forward Divided Difference Method. n is the number of data points. The loop checks if the value at which the solution is desired is between $x[1]$ and $x[n]$. If the value is between $x[1]$ and $x[n]$, then the slope from the closest points that bracket the value is found. The value of the slope is the value of the derivative at that point.

```

n := cols(x) - 1

Straight_Line := | for i ∈ 0..n - 1                                if x0,0 ≤ xv < x0,n-1
                  | AV ←  $\frac{y_{0,i+1} - y_{0,i}}{(x_{0,i+1} - x_{0,i})}$  if x0,i ≤ xv < x0,i+1
                  | "Point where derivative was requested is outside the domain of x" otherwise

```

The next method takes the three closest points to the given value to find a second order polynomial and differentiates it to find the value of $f'(x)$ at $x=xv$.

```

data1 :=  $\begin{pmatrix} x_{0,0} & y_{0,0} \\ x_{0,1} & y_{0,1} \\ x_{0,2} & y_{0,2} \end{pmatrix}$       data2 :=  $\begin{pmatrix} x_{0,n-2} & y_{0,n-2} \\ x_{0,n-1} & y_{0,n-1} \\ x_{0,n} & y_{0,n} \end{pmatrix}$ 

```

```

data3 := | for i ∈ 0..n - 2                                        if x0,1 ≤ xv < x0,n
          | | a ← i
          | | for j ∈ 0..2
          | | | Bj,0 ← x0,i if |x0,a+2 - xv| ≤ |xv - x0,a-1|
          | | | Bj,0 ← x0,i-1 otherwise
          | | | Bj,1 ← y0,i if (|x0,a+2 - xv| ≤ |xv - x0,a-1|)
          | | | Bj,1 ← y0,i-1 otherwise
          | | | i ← i + 1
          | | augment(B)

```

```

data := | A ← data1 if x0,0 ≤ xv < x0,1
        | A ← data2 if x0,n-1 ≤ xv < x0,n
        | A ← data3 otherwise
        | A

```

data := csort(data, 0) X := data^{<0>} Y := data^{<1>}

Spline coefficients:

S := lspline(X, Y)

Fitting function:

fit(a) := interp(S, X, Y, a)

dfit(x) := $\frac{d}{dx}$ fit(x)

dfit(xv) = 22.8

Next, all data points are going to be used to find a $n-1$ order polynomial. At the end, the polynomial is going to be differentiated and the value at which the derivative is wished to be found is going to be found.

datainterp := $\left| \begin{array}{l} \text{for } i \in 0..n \\ \left| \begin{array}{l} B_{i,0} \leftarrow x_{0,i} \\ B_{i,1} \leftarrow y_{0,i} \end{array} \right. \\ \text{augment}(B) \end{array} \right.$

datainterp := csort(datainterp, 0) X := datainterp^{<0>} Y := datainterp^{<1>}

Spline coefficients:

S2 := cspline(X, Y)

S3 := lspline(X, Y)

Fitting function:

fit2(b) := interp(S2, X, Y, b)

fit3(b) := interp(S3, X, Y, b)

dfit2(x) := $\frac{d}{dx}$ fit2(x)

dfit2(xv) = 22

Section 3: Table of Values


The next table shows the approximate values of the derivative at the given point using Forward Divic Difference, second and $n-1$ order polynomials.

$$a := \begin{pmatrix} . & \text{"Straight Line"} & \text{"Second Order"} & \text{"n-1th Order"} \\ \text{"f' (x)"} & \text{Straight_Line} & \text{dfit(xv)} & \text{dfit2(xv)} \end{pmatrix}$$

	0	1	2	3
a = 0	0	"Straight Line"	"Second Order"	"n-1th Order"
1	"f' (x)"	25	22.8	22

Section 4: Graphs

The following graph shows the the discrete data, linear splines and the n -*1*th order polynomial.



000 λ^{-1} data

References --- Linear splines

Numerical Differentiation of Continuous Functions. See
1-Dim 1-Order Polynomial
<http://numericalmethods.eng.usf.edu/mws/gen/02dif/>

Questions

1. The thermal expansion coefficient of steel is a function of temperature. Find the rate of change of thermal expansion coefficient as a function of temperature at $T=200$ F. Is this rate of change at $T=200$ F more or less than that at $T=50$ F? Use Forward Divided Difference to answer this question.
2. The distance traveled by a rocket is given as a function of time

t, s	0	10	20	30	40
x, miles	0	16	28	39	53

Find the rocket velocity and acceleration at $t=25$ s using numerical differentiation. Use all three methods illustrated in the worksheet.

Conclusions

The more data points taken to obtain the first derivative of a discrete function, more accurate the approximate value is. However, the more data points taken may result in oscillatory behavior generally observed with higher order interpolation. See

http://numericalmethods.eng.usf.edu/mws/gen/05inp/mws_gen_inp_spe_higherorder.pdf

Legal Notice: The copyright for this application is owned by the author(s). Neither Mathcad nor the author are responsible for any errors contained within and are not liable for any damages resulting from the use of this material. This application is intended for non-commercial, non-profit use only. Contact the author for permission if you wish to use this application for-profit activities.

